

NAG Library Routine Document

G05SRF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05SRF generates a vector of pseudorandom numbers from a von Mises distribution with concentration parameter κ .

2 Specification

```
SUBROUTINE G05SRF (N, VK, STATE, X, IFAIL)
```

```
INTEGER N, STATE(*), IFAIL
```

```
REAL (KIND=nag_wp) VK, X(N)
```

3 Description

The von Mises distribution is a symmetric distribution used in the analysis of circular data. The PDF (probability density function) of this distribution on the circle with mean direction $\mu_0 = 0$ and concentration parameter κ , can be written as:

$$f(\theta) = \frac{e^{\kappa \cos \theta}}{2\pi I_0(\kappa)},$$

where θ is reduced modulo 2π so that $-\pi \leq \theta < \pi$ and $\kappa \geq 0$. For very small κ the distribution is almost the uniform distribution, whereas for $\kappa \rightarrow \infty$ all the probability is concentrated at one point.

The n variates, $\theta_1, \theta_2, \dots, \theta_n$, are generated using an envelope rejection method with a wrapped Cauchy target distribution as proposed by Best and Fisher (1979) and described by Dagpunar (1988).

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SRF.

4 References

Best D J and Fisher N I (1979) Efficient simulation of the von Mises distribution *Appl. Statist.* **28** 152–157

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Mardia K V (1972) *Statistics of Directional Data* Academic Press

5 Parameters

1: N – INTEGER *Input*

On entry: n , the number of pseudorandom numbers to be generated.

Constraint: $N \geq 0$.

2: VK – REAL (KIND=nag_wp) *Input*

On entry: κ , the concentration parameter of the required von Mises distribution.

Constraint: $0.0 < VK \leq \sqrt{X02ALF}/2.0$.

- 3: STATE(*) – INTEGER array *Communication Array*
Note: the actual argument supplied must be the array STATE supplied to the initialization routines G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 4: X(N) – REAL (KIND=nag_wp) array *Output*
On exit: the n pseudorandom numbers from the specified von Mises distribution.
- 5: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $VK \leq 0.0$,
 or $VK > \sqrt{X02ALF()}/2.0$.

IFAIL = 3

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

For a given number of random variates the generation time increases slightly with increasing κ .

9 Example

This example prints the first five pseudorandom numbers from a von Mises distribution with $\kappa = 1.0$, generated by a single call to G05SRF, after initialization by G05KFF.

9.1 Program Text

```

Program g05srfe

!      G05SRF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: g05kff, g05srf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: vk
      Integer                    :: genid, ifail, lstate, n, subid
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: x(:)
      Integer                    :: seed(lseed)
      Integer, Allocatable        :: state(:)
!      .. Executable Statements ..
      Write (nout,*) 'G05SRF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!      Initial call to initialiser to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size
      Read (nin,*) n

      Allocate (x(n))

!      Read in the distribution parameters
      Read (nin,*) vk

!      Generate the variates
      ifail = 0
      Call g05srf(n,vk,state,x,ifail)

!      Display the variates
      Write (nout,99999) x(1:n)

99999 Format (1X,F10.4)
      End Program g05srfe

```

9.2 Program Data

```

G05SRF Example Program Data
1 1 1762543      :: GENID,SUBID,SEED(1)
5 3            :: N,NMIX
1.0           :: VK

```

9.3 Program Results

G05SRF Example Program Results

```
1.2947  
-1.9542  
-0.6464  
-1.4172  
1.2536
```
